Lösung zu Aufgabe (12)

Auszug aus einem Maple-Arbeitsblatt

Vorgegangen wird wie beim Beweis von Satz 13.2

```
> restart:with(linalg):
```

Die gegebene Matrix ist

```
> A:=matrix(3,2,[-2,4,28/10,4/10,38/10,-16/10]);
> #map(evalf,A);
```

$$A := \begin{bmatrix} -2 & 4 \\ \frac{14}{5} & \frac{2}{5} \\ \frac{19}{5} & \frac{-8}{5} \end{bmatrix}$$

Definition elementarer nxn-Drehungsmatrizen:

```
> eD:=proc(i,j,a,b,n) local M;
if i=j then ERROR("i darf nicht gleich j sein")
fi:
M:=evalm(array(1..n,1..n,identity));
M[i,i]:=a/sqrt(a^2+b^2):M[i,j]:=b/sqrt(a^2+b^2):M[j,i]:=-b/sqrt(a^2+b^2):M[j,j]:=a/sqrt(a^2+b^2):
evalm(M);
end:
```

1. Schritt: Berechnung eines invertierbaren quadratischen Blocks durch elementare orthogonale Umformungen.

```
> A0:=evalm(A):
> k:=1:
> A||k:=evalm(eD(1,2,A||(k-1)[1,1],A||(k-1)[2,1],3)&*A||(k-1));
U:=eD(1,2,A||(k-1)[1,1],A||(k-1)[2,1],3);
```

$$A1 := \begin{bmatrix} \frac{2\sqrt{74}}{5} & -\frac{43\sqrt{74}}{185} \\ 0 & -\frac{15\sqrt{74}}{37} \\ \frac{19}{5} & \frac{-8}{5} \end{bmatrix}$$

$$U := \begin{bmatrix} -\frac{5\sqrt{74}}{74} & \frac{7\sqrt{74}}{74} & 0 \\ -\frac{7\sqrt{74}}{74} & -\frac{5\sqrt{74}}{74} & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> k:=2:
> A||k:=map(simplify,evalm(eD(1,3,A||(k-1)[1,1],A||(k-1)[3,1],3)&*A||(k-1)));
U:=evalm(eD(1,3,A||(k-1)[1,1],A||(k-1)[3,1],3)&*U);
```

$$A2 := \begin{bmatrix} \frac{3\sqrt{73}}{5} & -\frac{108\sqrt{73}}{365} \\ 0 & -\frac{15\sqrt{74}}{37} \\ 0 & \frac{15\sqrt{74}\sqrt{73}}{2701} \end{bmatrix}$$

$$U := \begin{bmatrix} -\frac{10\sqrt{73}}{219} & \frac{14\sqrt{73}}{219} & \frac{19\sqrt{73}}{219} \\ -\frac{7\sqrt{74}}{74} & -\frac{5\sqrt{74}}{74} & 0 \\ \frac{95\sqrt{74}\sqrt{73}}{16206} & -\frac{133\sqrt{74}\sqrt{73}}{16206} & \frac{2\sqrt{74}\sqrt{73}}{219} \end{bmatrix}$$

```
> k:=3:
> A||k:=map(simplify,evalm(eD(2,3,A||(k-1)[2,2],A||(k-1)[3,2],3)&*A||(k-1)));
U:=evalm(eD(2,3,A||(k-1)[2,2],A||(k-1)[3,2],3)&*U);
```

$$A3 := \begin{bmatrix} \frac{3\sqrt{73}}{5} & -\frac{108\sqrt{73}}{365} \\ 0 & \frac{30\sqrt{73}}{73} \\ 0 & 0 \end{bmatrix}$$

$$U := \begin{bmatrix} -\frac{10\sqrt{73}}{219} & \frac{14\sqrt{73}}{219} & \frac{19\sqrt{73}}{219} \\ \frac{22\sqrt{73}}{219} & \frac{13\sqrt{73}}{219} & \frac{2\sqrt{73}}{219} \\ \frac{-1}{3} & \frac{2}{3} & \frac{-2}{3} \end{bmatrix}$$

Probe:

```
> evalm(A3-U&*A);
```

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \\ 0 & 0 \end{bmatrix}$$

Die folgende Teilmatrix ist jetzt invertierbar

```
> S:=submatrix(A3,1..2,1..2);
```

$$S := \begin{bmatrix} \frac{3\sqrt{73}}{5} & -\frac{108\sqrt{73}}{365} \\ 0 & \frac{30\sqrt{73}}{73} \end{bmatrix}$$

Nun wird zuerst $B = \operatorname{transpose}(S)\,S$ **diagonalisiert:**

```
> B:=evalm(transpose(S)&*S);
```

$$B := \begin{bmatrix} \frac{657}{25} & \frac{-324}{25} \\ \frac{-324}{25} & \frac{468}{25} \end{bmatrix}$$

das charakteristische Polynom ist:

```
> charpoly(B,t);
> factor(%);
```

$$t^2 - 45\,t + 324$$

$$(t-9)\,(t-36)$$

Berechnung einer orthonormierten Eigenvektorbasis. Beachte: die Orthogonalität gilt automatisch, da B **symmetrisch ist.**

```
> eigenvectors(B):
> Q:=concat(vector([-4/3, 1])/norm(vector([-4/3, 1]),2),vector([1, 4/3])/norm(vector([1,
4/3]),2));
```

$$Q := \begin{bmatrix} \frac{-4}{5} & \frac{3}{5} \\ \frac{3}{5} & \frac{4}{5} \end{bmatrix}$$

Probe:

```
> evalm(transpose(Q)&*Q);
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Festlegung der orthogonalen Matrix P **, wie im Beweis von Satz 13.2:**

```
> P:=evalm(diag(6,3)^(-1)&*transpose(Q)&*transpose(S));
```

$$P := \begin{bmatrix} -\frac{8\sqrt{73}}{73} & \frac{3\sqrt{73}}{73} \\ \frac{3\sqrt{73}}{73} & \frac{8\sqrt{73}}{73} \end{bmatrix}$$

Probe:

```
> evalm(transpose(P)&*P);
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Die Singulärwertzerlegung von A ist jetzt:

```
> evalm(   (diag(P,1)&*U)   &*   A   &*   Q   );
```

$$\begin{bmatrix} 6 & 0 \\ 0 & 3 \\ 0 & 0 \end{bmatrix}$$

```
>
```